Maple and Math Courses

Here are examples of some typical problems from some 200 and 300 level Math courses at UBC, and how Maple might be used to help solve them.

Math 200

Let D be the region inside the polar curve $r = 1 - \cos(\theta)$ and above the x axis. Evaluate the following integral:

$$\iint_D \frac{y}{\sqrt{x^2 + y^2}}\, dA$$

```
> restart; with(Student[MultivariateCalculus]):
> plot(1-cos(theta),theta=0..Pi,coords=polar,scaling=
  constrained);
```




```
> f:=changecoords(y/sqrt(x^2+y^2),[x,y],polar,[r,theta]);
```

$$f := \frac{r\sin(\theta)}{\sqrt{r^2\cos(\theta)^2 + r^2\sin(\theta)^2}} \tag{1.1}$$

```
> MultiInt(f, r = 0 .. 1-cos(theta), theta = 0 .. Pi,
  coordinates = polar);
```

$$\frac{4}{3} \tag{1.2}$$

Math 215

Solve the following system of differential equations with initial conditions.

$$\boldsymbol{x}' = \begin{bmatrix} 3 & -2 \\ 4 & -1 \end{bmatrix} . \boldsymbol{x} \qquad \boldsymbol{x}(0) = \begin{bmatrix} 1 \\ 1 \end{bmatrix}$$

```
> restart;
  sys:= D(x1)(t)=3*x1(t)-2*x2(t), D(x2)(t)=4*x1(t)-x2(t);
```

$$sys := \mathrm{D}(x1)(t) = 3\,x1(t) - 2\,x2(t), \mathrm{D}(x2)(t) = 4\,x1(t) - x2(t) \tag{2.1}$$

```
> ics:= x1(0)=1,x2(0)=1;
```

$$ics := x1(0) = 1, x2(0) = 1 \tag{2.2}$$

```
> dsolve({sys,ics});
```

$$\{x1(t) = e^t\cos(2\,t), x2(t) = e^t(\cos(2\,t) + \sin(2\,t))\} \tag{2.3}$$

)

Math 217

Find the maximum and minimum values of $f(x, y, z) = 4\,x - 2\,y$ subject to the constraints $2\,z - x - y = 4$ and $x^2 + y^2 = 1$.

```
> restart;
  with(Student[MultivariateCalculus]):
  LagrangeMultipliers(4*x-2*y,[2*z-x-y-4,x^2+z^2-1],[x,y,z],
  output=detailed);
```

Math 220

Determine whether the following series converges or diverges.

$$\sum_{n=1}^{\infty} \frac{n}{(3\,n)^n}$$

```
> restart;
```

Maple doesn't really do proofs, which are the main focus of Math 220. Also there isn't a command to decide whether a series converges or diverges. But you can do the usual tests for convergence, e.g. ratio test.

```
> a:= n -> n/(3*n)^n;
```

$$a := n \rightarrow \frac{n}{(3\,n)^n} \tag{4.1}$$

```
> limit(a(n+1)/a(n),n=infinity);
```

$$0 \tag{4.2}$$

So: it converges.

Math 221

Consider the following linear system:
$x + 3\,y - 2\,z + 2\,w = 1,$
$y + z - 2\,w = 2,$
$x + 2\,y - 2\,z + a\,w = 0,$
$2\,x + 8\,y - z + w = b$

For which values of a and b, if any, does the system have:
(i) No solution? (ii) Exactly one solution?
(iii) Exactly two solutions? (iv) More than two solutions?

```
> restart;
  sys := {x + 3*y - 2*z + 2*w = 1,
  y + z - 2*w = 2,
  x + 2*y - 2*z + a*w = 0,
    2*x + 8*y - z + w = b};
```

$$sys := \{y + z - 2\,w = 2, x + 2\,y - 2\,z + a\,w = 0, x + 3\,y - 2\,z + 2\,w = 1, 2\,x + 8\,y - z + w = b\} \tag{5.1}$$

```
> with(LinearAlgebra):
> A,B:= GenerateMatrix(sys,[x,y,z,w]);
```

$$A, B := \begin{bmatrix} 0 & 1 & 1 & -2 \\ 1 & 2 & -2 & a \\ 1 & 3 & -2 & 2 \\ 2 & 8 & -1 & 1 \end{bmatrix}, \begin{bmatrix} 2 \\ 0 \\ 1 \\ b \end{bmatrix} \tag{5.2}$$

```
> solve(Determinant(A)=0,a);
```

$$5 \tag{5.3}$$

```
> eval(sys,a=5);
```

(5.4)

$$\{y + z - 2w = 2, x + 2y - 2z + 5w = 0, x + 3y - 2z + 2w = 1, 2x + 8y - z + w = b\} \quad (5.4)$$

```
> solve(%,{x,y,z,w,b});
```

$$\{b = 7, w = 1 - z, x = -13 + 13z, y = 4 - 3z, z = z\} \quad (5.5)$$

So: (i) $a = 5, b \neq 7$; (ii) $a \neq 5$; (iii) never; (iv) $a = 5, b = 7$

Math 223

Let

$$A = \begin{bmatrix} 0 & 2 & 1 \\ 2 & 3 & 2 \\ 1 & 2 & 0 \end{bmatrix}$$

Determine an orthonormal basis of eigenvectors.

```
> restart;
  with(LinearAlgebra):
  A:= Matrix(<<0,2,1>|<2,3,2>|<1,2,0>>,shape=symmetric);
```

$$A := \begin{bmatrix} 0 & 2 & 1 \\ 2 & 3 & 2 \\ 1 & 2 & 0 \end{bmatrix} \quad (6.1)$$

```
> E,V:=Eigenvectors(A);
```

$$E, V := \begin{bmatrix} 5 \\ -1 \\ -1 \end{bmatrix}, \begin{bmatrix} 1 & -2 & -1 \\ 2 & 1 & 0 \\ 1 & 0 & 1 \end{bmatrix} \quad (6.2)$$

```
> Q:= [Column(V,1..3)];
```

$$Q := \left[\begin{bmatrix} 1 \\ 2 \\ 1 \end{bmatrix}, \begin{bmatrix} -2 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} -1 \\ 0 \\ 1 \end{bmatrix}\right] \quad (6.3)$$

```
> Normalize~(Q,Euclidean);
```

$$\left[\begin{bmatrix} \frac{1}{6}\sqrt{6} \\ \frac{1}{3}\sqrt{6} \\ \frac{1}{6}\sqrt{6} \end{bmatrix}, \begin{bmatrix} -\frac{2}{5}\sqrt{5} \\ \frac{1}{5}\sqrt{5} \\ 0 \end{bmatrix}, \begin{bmatrix} -\frac{1}{2}\sqrt{2} \\ 0 \\ \frac{1}{2}\sqrt{2} \end{bmatrix}\right] \quad (6.4)$$

Math 226

Prove that the line given by the parametric equations
$x = 1 + 4t, y = 2 - t, z = -3t$, is
parallel to the plane $2x + 5y + z = 4$.

```
> restart;
  2*(1+4*t) + 5*(2-t) + (-3*t);
```

$$12 \tag{7.1}$$

Math 227

A bug is flying through space so that its coordinates at time t are $x(t) = (t^2 + t, t^2 - t, t^3)$. Find the bug's velocity and acceleration for all t, and the curvature of its trajectory at time $t = 0$.

```
> restart;
  X := <t^2 + t, t^2 - t, t^3>;
```

$$X := \begin{bmatrix} t^2 + t \\ t^2 - t \\ t^3 \end{bmatrix} \tag{8.1}$$

```
> V:= map(diff, X, t);
```

$$V := \begin{bmatrix} 2\,t + 1 \\ 2\,t - 1 \\ 3\,t^2 \end{bmatrix} \tag{8.2}$$

```
> A:= map(diff, X, t$2);
```

$$A := \begin{bmatrix} 2 \\ 2 \\ 6\,t \end{bmatrix} \tag{8.3}$$

```
> K:= eval(VectorCalculus[Curvature](X,t),t=0);
```

$$K := \frac{1}{2}\sqrt{4}\,\sqrt{2} \tag{8.4}$$

```
> simplify(%);
```

$$\sqrt{2} \tag{8.5}$$

Math 300

Find all complex solutions to the equation
$\cos(z) = 2\,i\,\sin(z)$.
Express each solution in the form $z = x + y\,i$, where x and y are real numbers.

```
> restart;
```

```
> solve(cos(z)=2*I*sin(z),z,AllSolutions);
```

$$-\mathrm{I}\,\mathrm{arctanh}\left(\frac{1}{2}\right)+\pi\_Z1\sim \quad \textbf{(9.1)}$$

```
> convert(%,ln);
```

$$-\mathrm{I}\left(\frac{1}{2}\ln\left(\frac{3}{2}\right)+\frac{1}{2}\ln(2)\right)+\pi\_Z1\sim \quad \textbf{(9.2)}$$

Math 301

Evaluate $\int_0^\infty \frac{x}{8+x^3}\,dx$

```
> restart;
  int(x/(8+x^3),x=0..infinity);
```

$$\frac{1}{9}\pi\sqrt{3} \quad \textbf{(10.1)}$$

Math 302

Let X and Y be independent standard normal random variables, and let $Z=X^2+Y^2$.
Find the cumulative distribution function of Z, and be sure to specify your answer for all values of z.

```
> restart;
  with(Statistics):
> X:= RandomVariable(Normal(0,1)): Y:= RandomVariable(Normal(0,
  1)):
  Z:= X^2 + Y^2;
```

$$Z:=\_R^2+\_R0^2 \quad \textbf{(11.1)}$$

```
> CDF(Z,z);
```

$$\begin{cases} 0 & z\leq 0 \\ -\mathrm{e}^{-\frac{1}{2}z}+1 & 0<z \end{cases} \quad \textbf{(11.2)}$$

Math 303

A Markov chain has state space $\{1, 2, 3, 4, 5, 6\}$ and transition matrix

$$\begin{bmatrix} 0 & 0 & \frac{3}{4} & 0 & \frac{1}{4} & 0 \\ 0 & \frac{2}{5} & 0 & 0 & \frac{1}{5} & \frac{2}{5} \\ 0 & \frac{2}{3} & 0 & \frac{1}{3} & 0 & 0 \\ \frac{2}{7} & 0 & \frac{5}{7} & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \end{bmatrix}$$

(a) Find the communicating classes.
(b) What is the probability, starting in 3, that the Markov chain never visits state 4?.

```
> restart;
  with(GraphTheory):
  P := Matrix([[0,0,3/4,0,1/4,0],
  [0,2/5,0,0,1/5,2/5],
  [0,2/3,0,1/3,0,0],
  [2/7,0,5/7,0,0,0],
  [0,0,0,0,1,0],
  [0,1,0,0,0,0]]);
  G:= Digraph(P);
  DrawGraph(G);
  Classes:= StronglyConnectedComponents(G);
```

$$P := \begin{bmatrix} 0 & 0 & \frac{3}{4} & 0 & \frac{1}{4} & 0 \\ 0 & \frac{2}{5} & 0 & 0 & \frac{1}{5} & \frac{2}{5} \\ 0 & \frac{2}{3} & 0 & \frac{1}{3} & 0 & 0 \\ \frac{2}{7} & 0 & \frac{5}{7} & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \end{bmatrix}$$

G := Graph 1: a directed weighted graph with 6 vertices and 11 arc(s)



$$Classes := [[5], [2, 6], [1, 3, 4]] \tag{12.1}$$

```
> eqs:= {seq(u[j] = add(P[j,k]*u[k],k=[1,3,4])+add(P[j,k],k=[5,
  2,6]),j=[1,3]),
  u[4]=0};
```

$$eqs := \left\{u_1 = \frac{3}{4}\ u_3 + \frac{1}{4}, u_3 = \frac{1}{3}\ u_4 + \frac{2}{3}, u_4 = 0\right\} \tag{12.2}$$

```
> solve(eqs);
```

$$\left\{u_1 = \frac{3}{4}, u_3 = \frac{2}{3}, u_4 = 0\right\} \tag{12.3}$$

Math 307

Let V denote the row space of the matrix

$$\begin{bmatrix} 1 & 1 & 0 \\ 2 & 0 & 1 \\ 1 & 5 & -2 \end{bmatrix}$$

Find bases of V and $V^{\perp}$.

```
> restart;
  with(LinearAlgebra):
  A:= <<1,2,1>|<1,0,5>|<0,1,-2>>;
```

$$A := \begin{bmatrix} 1 & 1 & 0 \\ 2 & 0 & 1 \\ 1 & 5 & -2 \end{bmatrix} \quad \textbf{(13.1)}$$

```
> RowSpace(A);
```

$$\left[\left[\begin{matrix} 1 & 0 & \frac{1}{2} \end{matrix} \right], \left[\begin{matrix} 0 & 1 & -\frac{1}{2} \end{matrix} \right] \right] \quad \textbf{(13.2)}$$

```
> NullSpace(A);
```

$$\left\{ \begin{bmatrix} -\frac{1}{2} \\ \frac{1}{2} \\ 1 \end{bmatrix} \right\} \quad \textbf{(13.3)}$$

Math 308

Consider the triangle ABC in $\mathbb{R}^2$ whose vertices A, B, C have coordinates $(0, 1)$, $(t, 0)$ and $(s, 0)$, respectively.
Find the coordinates of the centroid G of the orthocentre H.

```
> restart; with(geometry):
  assume(t <> s):
  triangle(ABC,[point(A,0,1),point(B,t,0),point(C,s,0)]);
```

$$ABC \quad \textbf{(14.1)}$$

```
> coordinates(centroid(G,ABC));
```

$$\left[\frac{1}{3} t\sim + \frac{1}{3} s\sim, \frac{1}{3} \right] \quad \textbf{(14.2)}$$

```
> coordinates(orthocenter(H,ABC));
```

$$\left[0, \frac{t\sim s\sim^2 - t\sim^2 s\sim}{-s\sim + t\sim} \right] \quad \textbf{(14.3)}$$

Math 312

Let n be a four-digit integer $a_0 + a_1 \cdot 10 + a_2 \cdot 10^2 + a_3 \cdot 10^3$. Show that n is divisible by 7 if and only if $a_0 + 3\,a_1 + 2\,a_2 - a_3$ is divisible by 7.

```
> restart;
  n:= add(a[j]*10^j, j=0..3);
```

$$n := a_0 + 10\, a_1 + 100\, a_2 + 1000\, a_3 \quad \textbf{(15.1)}$$

```
> n mod 7, a[0] + 3*a[1] + 2*a[2] - a[3] mod 7;
```

$$a_0 + 3\, a_1 + 2\, a_2 + 6\, a_3,\ a_0 + 3\, a_1 + 2\, a_2 + 6\, a_3 \quad \textbf{(15.2)}$$

Math 313

Find positive integers x and y such that $x^2 + y^2 = 5 \cdot 13 \cdot 17 \cdot 29$

```
> restart;
  isolve(x^2 + y^2 = 5*13*17*29);
```

$\{x=-179, y=-2\}, \{x=-179, y=2\}, \{x=-178, y=-19\}, \{x=-178, y=19\}, \{x=-173, y=-46\}, \{x=-173, y=46\}, \{x=-166, y=-67\}, \{x=-166, y=67\}, \{x=-163, y=-74\}, \{x=-163, y=74\}, \{x=-157, y=-86\}, \{x=-157, y=86\}, \{x=-142, y=-109\}, \{x=-142, y=109\}, \{x=-131, y=-122\}, \{x=-131, y=122\}, \{x=-122, y=-131\}, \{x=-122, y=131\}, \{x=-109, y=-142\}, \{x=-109, y=142\}, \{x=-86, y=-157\}, \{x=-86, y=157\}, \{x=-74, y=-163\}, \{x=-74, y=163\}, \{x=-67, y=-166\}, \{x=-67, y=166\}, \{x=-46, y=-173\}, \{x=-46, y=173\}, \{x=-19, y=-178\}, \{x=-19, y=178\}, \{x=-2, y=-179\}, \{x=-2, y=179\}, \{x=2, y=-179\}, \{x=2, y=179\}, \{x=19, y=-178\}, \{x=19, y=178\}, \{x=46, y=-173\}, \{x=46, y=173\}, \{x=67, y=-166\}, \{x=67, y=166\}, \{x=74, y=-163\}, \{x=74, y=163\}, \{x=86, y=-157\}, \{x=86, y=157\}, \{x=109, y=-142\}, \{x=109, y=142\}, \{x=122, y=-131\}, \{x=122, y=131\}, \{x=131, y=-122\}, \{x=131, y=122\}, \{x=142, y=-109\}, \{x=142, y=109\}, \{x=157, y=-86\}, \{x=157, y=86\}, \{x=163, y=-74\}, \{x=163, y=74\}, \{x=166, y=-67\}, \{x=166, y=67\}, \{x=173, y=-46\}, \{x=173, y=46\}, \{x=178, y=-19\}, \{x=178, y=19\}, \{x=179, y=-2\}, \{x=179, y=2\}$ **(16.1)**

```
> remove(hastype,[%],negative);
```

$[\{x=2, y=179\}, \{x=19, y=178\}, \{x=46, y=173\}, \{x=67, y=166\}, \{x=74, y=163\}, \{x=86, y=157\}, \{x=109, y=142\}, \{x=122, y=131\}, \{x=131, y=122\}, \{x=142, y=109\}, \{x=157, y=86\}, \{x=163, y=74\}, \{x=166, y=67\}, \{x=173, y=46\}, \{x=178, y=19\}, \{x=179, y=2\}]$ **(16.2)**

Math 316

Consider the differential equation $(x^3 - 1)\, y'' + x\, y' + 2\, y = 0$. Show that $x = 1$ is a regular singular point, and find the indicial equation at this point.

```
> restart;
  de:= (x^3 - 1)*diff(y(x),x$2) + x*diff(y(x),x) + 2*y(x)=0;
```

(17.1)

$$de := (x^3 - 1)\left(\frac{d^2}{dx^2}\,y(x)\right) + x\left(\frac{d}{dx}\,y(x)\right) + 2\,y(x) = 0 \qquad \textbf{(17.1)}$$

```
> with(DEtools):
  singularities(de);
```

$$regular = \left\{1,\ \infty,\ -\frac{1}{2} - \frac{1}{2}\,I\sqrt{3},\ -\frac{1}{2} + \frac{1}{2}\,I\sqrt{3}\right\},\ irregular = \{\ \} \qquad \textbf{(17.2)}$$

```
> indicialeq(de,r,1,y(x));
```

$$r^2 - r = 0 \qquad \textbf{(17.3)}$$

Math 317

Evaluate $\iint_S (\text{curl}\,\mathbf{F}\,.\,\mathbf{n})\,dS$ where S is that part of the sphere $x^2 + y^2 + z^2 = 2$ above the plane $z = 1$, $\mathbf{n}$ is the upward unit normal, and
$\mathbf{F}(x, y, z) = y^2\,\mathbf{i} + x^3\,\mathbf{j} + (e^x + e^y + z)\,\mathbf{k}$

```
> restart;
  with(VectorCalculus):
  SetCoordinates(cartesian[x,y,z]):
  S:= Surface(<x,y,sqrt(2-x^2-y^2)>,[x,y]=Circle(<0,0>,sqrt(2))
  ):
  F:= VectorField(<y^2,x^3,exp(x)+exp(y)+z>):
  Flux(Curl(F),S);
```

$$3\,\pi \qquad \textbf{(18.1)}$$

Math 318

A closet contains 10 different pairs of shoes (each pair consists of a left shoe and a right shoe, so there are 20 shoes in total). 6 shoes are chosen at random. Find the probability that exactly one complete pair is chosen.

```
> restart;
  with(combinat):
  shoes:= {seq(right[i],i=1..10),seq(left[i],i=1..10)};
```

$$shoes := \{left_1, left_2, left_3, left_4, left_5, left_6, left_7, left_8, left_9, left_{10}, right_1, right_2, right_3, right_4, right_5, right_6, right_7, right_8, right_9, right_{10}\} \qquad \textbf{(19.1)}$$

```
> SampleSpace:= choose(shoes,6):
> SampleSpace[1..3];
```

$$\{\{left_1, left_2, left_3, left_4, left_5, left_6\}, \{left_1, left_2, left_3, left_4, left_5, left_7\}, \{left_1, left_2, left_3, left_4, left_5, left_8\}\} \qquad \textbf{(19.2)}$$

```
> countpairs:= w -> add(piecewise({left[j],right[j]} subset w,
```

```
1, 0),j=1..10);
```

$$countpairs := w \rightarrow add\left(piecewise\left(\{left_j, right_j\} \subseteq w, 1, 0\right), j = 1..10\right) \quad (19.3)$$

```
> nops(select(t -> (countpairs(t)=1),SampleSpace))/nops
(SampleSpace);
```

$$\frac{168}{323} \quad (19.4)$$

```
> binomial(6,2)* 1/19 * 18/18 * 16/17 * 14/16 * 12/15;
```

$$\frac{168}{323} \quad (19.5)$$

Math 321

Evaluate $\sum_{n=-\infty}^{\infty} \left| \int_{-\pi}^{\pi} t^5 e^{-int} \, dt \right|^2$.

```
> restart;
J:= int(t^5*exp(-I*n*t),t=-Pi..Pi) assuming n::integer;
```

$$J := \frac{\left(240 \, I \, n \, \pi - 40 \, I \, \pi^3 \, n^3 + 2 \, I \, \pi^5 \, n^5\right) (-1)^n}{n^6} \quad (20.1)$$

```
> J2:=simplify(abs(J)^2) assuming n::posint;
```

$$J2 := \frac{4 \, \pi^2 \left(120 - 20 \, n^2 \, \pi^2 + n^4 \, \pi^4\right)^2}{n^{10}} \quad (20.2)$$

The case $n = 0$ must be done separately.

```
> int(t^5,t=-Pi..Pi);
```

$$0 \quad (20.3)$$

```
> 2*sum(J2,n=1..infinity);
```

$$\frac{4}{11} \pi^{12} \quad (20.4)$$

Math 322

Factor the following elements of the given rings R into irreducible elements of R

(a) 15, an element of $R = \{a + b\sqrt{-2}\}$

```
> numtheory[factorEQ](15,-2);
```

$$\left(1 + I\sqrt{2}\right) \left(1 - I\sqrt{2}\right) (5) \quad (21.1)$$

(b) $5\,x^4 - 20\,x^3 + 30$, an element of $R = Z[x]$

```
> factor(5*x^4 - 20*x^3 + 30);
```

(21.2)

$$5\,x^4 - 20\,x^3 + 30 \quad \textbf{(21.2)}$$

(c) $x^3 + 1$, an element of $R = \mathbb{F}_7\,[x]$

```
> Factor(x^3+1) mod 7;
```

$$(x+2)\,(x+4)\,(x+1) \quad \textbf{(21.3)}$$

Math 340

Solve the linear programming problem

maximize $x_1 + x_2 + x_3$

subject to $x_1 + 2\,x_2 + 3\,x_3 \le 6$,

$2\,x_1 - 2\,x_2 + x_3 \ge 5$,

$x_1 - x_2 + x_3 = 4$,

$x_1, x_2, x_3 \ge 0$

```
> restart;
  with(simplex):
  maximize(x[1]+x[2]+x[3],
  {x[1]+2*x[2]+3*x[3]<=6,
   2*x[1]-2*x[2]+x[3]>=5,
   x[1]-x[2]+x[3]=4},NONNEGATIVE);
```

$$\left\{x_1 = \frac{14}{3},\, x_2 = \frac{2}{3},\, x_3 = 0\right\} \quad \textbf{(22.1)}$$

Math 342

Show that S_3 is generated by the cycles (1,2) and (2,3).

```
> restart;
  with(group):
> permgroup(3,{[[1,2]],[[2,3]]});
```

$$permgroup(3,\, \{[[1, 2]],\, [[2, 3]]\}) \quad \textbf{(23.1)}$$

```
> grouporder(%);
```

$$6 \quad \textbf{(23.2)}$$

Math 345

Find a conserved quantity for the two-dimensional system

$\dot{x} = y$,

$\dot{y} = x^2 - 4$

```
> restart;
  sys := {D(x)(t)=y(t), D(y)(t)=x(t)^2-4};
```

$$sys := \{\mathrm{D}(x)(t) = y(t),\, \mathrm{D}(y)(t) = x(t)^2 - 4\} \quad \textbf{(24.1)}$$

```
> with(DEtools):
> casesplit(sys);
```

$$\left[y(t) = \frac{d}{dt}\,x(t),\ \frac{d^2}{dt^2}\,x(t) = x(t)^2 - 4\right] \&\text{where } [\ ] \qquad \textbf{(24.2)}$$

```
> xde:= op([1,2],%);
```

$$xde := \frac{d^2}{dt^2}\,x(t) = x(t)^2 - 4 \qquad \textbf{(24.3)}$$

```
> mu:= intfactor(xde);
```

$$\mu := \frac{d}{dt}\,x(t) \qquad \textbf{(24.4)}$$

```
> firint(xde*mu);
```

$$-\frac{2}{3}\,x(t)^3 + 8\,x(t) + \left(\frac{d}{dt}\,x(t)\right)^2 + \_C1 = 0 \qquad \textbf{(24.5)}$$

```
> solve(subs(diff(x(t),t)=y(t),%),_C1);
```

$$\frac{2}{3}\,x(t)^3 - 8\,x(t) - y(t)^2 \qquad \textbf{(24.6)}$$

Math 361

Schnakenberg (1979) considered the following simplified model of glycolysis:

$$\frac{dx}{dt} = x^2 y - x$$

$$\frac{dy}{dt} = a - x^2 y$$

where $a > 0$. As the parameter a varies, the steady state of the system changes its behaviour. Determine the sequence of steady state classifications for increasing values of a (e.g. stable node-to-saddle-
to-unstable node). Does the system undergo a Hopf bifurcation for any value(s) of a? If so, at what value(s) does the Hopf bifurcation(s) occur?

```
> restart;
  F:= [x^2*y - x, a - x^2*y];
```

$$F := [x^2 y - x,\ a - x^2 y] \qquad \textbf{(25.1)}$$

```
> S:=solve(F,{x,y});
```

$$S := \left\{x = a,\ y = \frac{1}{a}\right\} \qquad \textbf{(25.2)}$$

```
> M:= eval(VectorCalculus[Jacobian](F,[x,y]),S);
```

$$M := \begin{bmatrix} 1 & a^2 \\ -2 & -a^2 \end{bmatrix} \quad (25.3)$$

```
> with(LinearAlgebra):
> solve({discrim(CharacteristicPolynomial(M,t),t)=0,a>0});
```

$$\{a = \sqrt{2} - 1\}, \{a = 1 + \sqrt{2}\} \quad (25.4)$$

```
> solve({Trace(M)=0,a>0});
```

$$\{a = 1\} \quad (25.5)$$

```
> Eigenvalues(eval(M,a=0.1));
```

$$\begin{bmatrix} 0.979793770587041 + 0.\,I \\ 0.0102062294129596 + 0.\,I \end{bmatrix} \quad (25.6)$$

```
> Eigenvalues(eval(M,a=sqrt(2)-1));
```

$$\begin{bmatrix} \sqrt{2} - 1 \\ \sqrt{2} - 1 \end{bmatrix} \quad (25.7)$$

```
> Eigenvalues(eval(M,a=0.5));
```

$$\begin{bmatrix} 0.375000000000000 + 0.330718913883074\,I \\ 0.375000000000000 - 0.330718913883074\,I \end{bmatrix} \quad (25.8)$$

```
> Eigenvalues(eval(M,a=1));
```

$$\begin{bmatrix} I \\ -I \end{bmatrix} \quad (25.9)$$

```
> Eigenvalues(eval(M,a=1.1));
```

$$\begin{bmatrix} -0.105000000000000 + 1.09497716871175\,I \\ -0.105000000000000 - 1.09497716871175\,I \end{bmatrix} \quad (25.10)$$

```
> Eigenvalues(eval(M,a=sqrt(2)+1));
```

$$\begin{bmatrix} -1 - \sqrt{2} \\ -1 - \sqrt{2} \end{bmatrix} \quad (25.11)$$

```
> Eigenvalues(eval(M,a=2.5));
```

$$\begin{bmatrix} -1.82460947032089 + 0.\,I \\ -3.42539052967911 + 0.\,I \end{bmatrix} \quad (25.12)$$

Conclusion: unstable node to unstable spiral at $a = \sqrt{2} - 1$, to stable spiral at $a = 1$, to stable node at $a = \sqrt{2} + 1$. Hopf bifurcation occurs at $a = 1$.